

```

ctl-opt pgminfo(*pcml:*module) nomain debug
        option( *srcstmt : *nodebugio );

//      This program receives an item class and returns the first 200
//      items found in the item master table.

//      Note - If you wanted to have a 'conversation' and continue
//      returning each subsequent set of 200 you would need to
//      receive the last item sent to the consumer and use
//      that as your starting point.

//      Data structure webordersdata's only purpose is to define table fields.
dcl-ds webordersdata extname('WEBITEMS') template end-ds;

//      If you want to return more than 200 items, just change the value here.
dcl-c maxelements const(200);

dcl-ds itemlistds qualified;
        itemnumber like(acitemno);
        description like(acitemdes);
        unitofmeasure like(acunitofms);
        itemclass like(acitemclas);
end-ds;

//      Subprocedure getitems starts here.
//      Since this is a local subprocedure we don't need a prototype.

dcl-proc getitems export;

dcl-pi *n;
itemclassin char(2) const;
success char(5);
classdesc char(30);
numberofitems int(10);
itemlist likeds(itemlistds)
        dim(maxelements);
end-pi;

dcl-s maxfetch packed(3:0) inz(maxelements);

// Reset important indicator fields, this program is stateless
numberofitems = *zero;
success = 'False';

// Retrieve item class description from master table
exec sql
        select ahclasdesc into :classdesc
        from webclsdes
        where ahitemclas = :itemclassin;

```

```
// Prepare to read and return items in requested item class
exec sql
  declare c1 cursor for
    select acitemno,
           acitemdes,
           acunitofms,
           acitemclas
    from webitems
    where acitemclas = :itemclassin
    order by acitemno;

exec sql
  open c1;

// Up to 200 rows will be placed into data structure array itemlist
exec sql
  fetch c1 for :maxfetch rows into :itemlist;

// Returns the number of rows
// row_count comes from the SQL communications area
exec sql
  get diagnostics :numberofitems = row_count;

exec sql
  close c1;

// If any items were returned set success flag to "True"
if numberofitems > *zero;
  success = 'True';
endif;

  return;

end-proc getitems;
```